Module 11 Polynomen en rationale functies

Onderwerp	Manipulaties met polynomen en rationale functies.
Expressies	`factor, collect, coeff, degree, lcoeff, gcd, numer, denom, normal, algsubs, convert( ,parfrac)`
Zie ook	Module 4, 12, 26, 15

11.1 Polynomen en rationale functies

Een *polynoom* in de variabelen $x_1, \ldots, x_n$ is een expressie die bestaat uit sommen en producten van deze variabelen. Een *rationale functie* is het quotiënt van twee polynomen.

11.2 Factoriseren van polynomen in één variabele

Voor het factoriseren van polynomen is het commando `factor` bedoeld. Hiermee hebben we in Module 4 al kennisgemaakt. In veel gevallen zal `factor(p)` niet het gewenste resultaat geven, zelfs als p een polynoom van de graad 2 is met twee reëele nulpunten. Dat komt omdat Maple in een eventuele factorisatie uitsluitend gebruik zal maken van het 'soort getallen' dat ook in de coëfficiënten is gebruikt. Preciezer: Als alle coëfficiënten van p gehele getallen zijn, dan zal Maple een factorisering proberen te vinden waarin uitsluitend rationale getallen voorkomen.

`factor`

Het commando `factor` kan echter ook met een tweede argument worden aangeroepen. Dit argument bestaat uit

- een verzameling die alle overige getallen bevat die ook gebruikt mogen worden, óf
- het woord `real` of `complex`.

Zie `?factor` voor een nauwkeuriger beschrijving en de volgende sessie voor een paar voorbeelden.

Voorbeeldsessie

```
> restart; Digits := 5:
```

Een polynoom met niet-rationale nulpunten

```
> p1 := x^2-x-1;
```

$$p1 := x^2 - x - 1$$

```
> factor(p1);
```

$$x^2 - x - 1$$

```
> factor(p1,{sqrt(5)});
```

$$\frac{(2\,x - 1 + \sqrt{5})\,(2\,x - 1 - \sqrt{5})}{4}$$

Dezelfde polynoom met een van de coëfficiënten als float genoteerd:

```
> p2 := x^2-x-1.0;
```

$$p2 := x^2 - x - 1.0$$

```
> factor(p2);
```

$$(x + 0.61803)\,(x - 1.6180)$$

Hetzelfde effect wordt bereikt met:

```
> factor(p1,real);
```

$$(x + 0.61803)\,(x - 1.6180)$$

Vierdegraadspolynoom met reële en complexe nulpunten:

```
> p3 := x^4-x^2-2*x-1;
```

$$p3 := x^4 - x^2 - 2\,x - 1$$

```
> factor(p3);
```

$$(x^2 - x - 1)\,(x^2 + x + 1)$$

```
> factor(p3,{sqrt(5)});
```

$$\frac{(x^2 + x + 1)\,(2\,x - 1 + \sqrt{5})\,(2\,x - 1 - \sqrt{5})}{4}$$

```
> factor(p3,{I,sqrt(3),sqrt(5)});
```

$$(2\,x - 1 + \sqrt{5})\,(2\,x - 1 - \sqrt{5})\,(2\,x + 1 + \sqrt{3}\,I)$$
$$(2\,x + 1 - \sqrt{3}\,I)/16$$

```
> factor(p3,real);
```

$$(x + 0.61803)\,(x - 1.6180)\,(x^2 + 1.0000\,x + 1.0000)$$

```
> factor(p3,complex);
```

$$(x + 0.61803)\,(x + 0.50000 + 0.86603\,I)$$
$$(x + 0.50000 - 0.86603\,I)\,(x - 1.6180)$$

Toelichting

Van `p1` bestaat geen factorisatie met rationale coëfficiënten. Als we Maple vertellen dat in de ontbinding van `p1` ook het getal $\sqrt{5}$ mag worden gebruikt[31], wordt inderdaad de gewenste factorisatie berekend.

[31] Dit hebben we bedacht nadat we met `solve` de nulpunten van `p1` hebben bepaald

In de polynoom `p2` komt de coëfficiënt 1.0 voor, en dit is voor Maple een signaal dat bij alle berekeningen decimale benaderingen moeten worden gemaakt.
Van `p3` wordt aanvankelijk slechts de factorisatie in tweedegraads factoren gegeven. Door toevoeging van de juiste getallen in het tweede argument wordt respectievelijk de reële en de complexe factorisatie gegeven.
Let op dat `factor(p3,complex)` een andere betekenis heeft dan `factor(p3,{I})`. In het eerste geval maakt Maple *decimale benaderingen* van de complexe factoren, terwijl in het tweede geval geprobeerd wordt een *exacte* complexe ontbinding (zonder wortels) te maken. ◇

11.3 Herschrijfregels

`collect`

In §4.2 zagen we al dat we door middel van `expand` en `factor` de gedaante van polynomen kunnen beïnvloeden. Voor polynomen in meer dan één variabele is er bovendien nog een andere belangrijke opdracht, namelijk `collect`. Met deze opdracht kan men opgeven welke variabele als belangrijkste wordt opgevat.
We kunnen bijvoorbeeld de polynoom

$$p = x^3 + xy + xy^2 + x^2y^2 \tag{11.1}$$

schrijven als

$$p_1 = x(y + y^2) + x^2y^2 + x^3. \tag{11.2}$$

Dan vatten we p_1 op als een polynoom in de variabele x, met coefficiënten die functies zijn van y. Wanneer we echter p schrijven als

$$p_2 = x^3 + yx + y^2(x + x^2), \tag{11.3}$$

dan vatten we p_2 juist op als polynoom in de variabele y, met coefficiënten die functies zijn van x. Voor Maple zijn beide polynomen niet automatisch gelijk. (Dat komt omdat hun interne representatie (GAG) verschillend is, zie Module 26.)

Voorbeeldopgave

Herschrijf p uit (11.1) zowel in de vorm p_1 uit (11.2) als in de vorm p_2 uit (11.3). Bereken $p_1 - p_2$.

Voorbeeldsessie

```
> p := x^3 + x*y + x*y^2 + x^2*y^2;
```

$$p := x^3 + x\,y + x\,y^2 + x^2\,y^2$$

```
> p1 := collect(p,x);  p2 := collect(p,y);
```

$$p1 := x^3 + x^2\,y^2 + (y + y^2)\,x$$

$$p2 := (x + x^2)\,y^2 + x\,y + x^3$$

```
> p1-p2;
```

$$x^2\,y^2 + (y + y^2)\,x - (x + x^2)\,y^2 - x\,y$$

```
> simplify(%);
```

$$0$$

Toelichting

Pas na het gebruik van `simplify` levert Maple 0 op voor $p_1 - p_2$. ◇

11.4 Coëfficiënten van polynomen

Wanneer een polynoom p van meer dan twee variabelen afhangt, kunnen we, na een aanroep van `collect`, de coëfficiënten van p weer opvatten als polynomen in meer dan één variabele. Het kan voorkomen dat we hierop weer `collect` willen toepassen, maar `collect` werkt op de hele polynoom, dus dat gaat niet. Echter, we kunnen in `collect` wel meer dan één variabele opgeven. Dit geeft dan de 'orde van belangrijkheid' van de variabelen volgens welke de termen in p gegroepeerd moeten worden. Bijvoorbeeld de polynoom

$$p = xyz + xy^2 + xy^2z + xz + xz^2 + x^2y + x^2zy + x^2 + yz$$

kunnen we op diverse manieren herschrijven:

Voorbeeldsessie

```
> p := x*y*z   + x*y^2 + x*y^2*z +
       x*z     + x*z^2 + x^2*y   +
       x^2*z*y + x^2   + y*z;
```

$$p := x\,y\,z + x\,y^2 + x\,y^2\,z + x\,z + x\,z^2 + x^2\,y + x^2\,z\,y + x^2 + y\,z$$

```
> p1 := collect( p, [x,y], recursive );
```

$$p1 := (1 + (1 + z)\,y)\,x^2 + ((1 + z)\,y^2 + y\,z + z^2 + z)\,x + y\,z$$

```
> p2 := collect( p, [y,x], recursive );
```

$$p2 := (1 + z)\,x\,y^2 + ((1 + z)\,x^2 + x\,z + z)\,y + x^2 + (z^2 + z)\,x$$

```
> p3 := collect( p, [y,x], distributed );
```

$$p3 := x\,y\,z + x^2 + (1+z)\,x\,y^2 + (1+z)\,x^2\,y + y\,z + (z^2+z)\,x$$

Toelichting

Het verschil is dat in `p1` de variabele `x` als belangrijkste variabele wordt opgevat, en `y` hieraan ondergeschikt is: eerst vindt groepering van de termen plaats naar machten van `x`, en binnen de aldus ontstane coëfficiënten wordt gegroepeerd naar machten van `y`. In `p2` is de rol van `x` en `y` verwisseld. In `p3` worden de variabelen `x` en `y` gelijkwaardig beschouwd, en vindt groepering van de termen plaats naar termen van de vorm $x^m y^n$. ◇

`coeff` `algsubs`

Na een aanroep van `collect` kunnen we de coëfficiënten van `p` in principe opvragen met `coeff`. Bijvoorbeeld de coëfficient van x^2 in de polynoom p vinden we met `coeff(p, x, 2)`. Echter, wanneer we in p de coëfficient van x^2y willen weten dan werkt `coeff` niet. Dat komt omdat x^2y niet een operande is van p (zie Module 26). We moeten dan eerst x^2y vervangen door iets anders (via `algsubs`), waarna we de coëfficiënt kunnen bereiken. Een andere mogelijkheid wordt behandeld in §12.3.

Voorbeeldsessie

```
> p := x*y*z   + x*y^2 + x*y^2*z +
       x*z     + x*z^2 + x^2*y   +
       x^2*z*y + x^2   + y*z;
```

$$p := x\,y\,z + x\,y^2 + x\,y^2\,z + x\,z + x\,z^2 + x^2\,y + x^2\,z\,y + x^2 + y\,z$$

```
> coeff(p, x, 2);
```

$$y + y\,z + 1$$

De coëfficiënt van $x\,y^2$ is niet zonder meer te verkrijgen met coeff:

```
> c3 := coeff(p, x*y^2);

Error, invalid input: coeff received x*y^2, which is not valid
for its 2nd argument, x
```

... want $x\,y^2$ is geen deelexpressie van p.

```
> pnieuw := algsubs( x*y^2=XY2, p );
```

$$pnieuw := y\,z + (z+1)\,x^2\,y + x\,y\,z + XY2 + z\,XY2 + x^2 + (z^2+z)\,x$$

```
> c3 := coeff(pnieuw, XY2);
```

$$c3 := z + 1$$

Andere mogelijkheden. Er zijn nog meer commando’s die behulpzaam kunnen zijn bij het manipuleren met polynomen.

`degree`

We noemen `degree(p,x)` dat de *graad* aflevert van `p`, opgevat als

polynoom in `x`.
Verder is er het commando `lcoeff` ('leading coefficient').

`lcoeff` Als $\mathrm{grd}(p) = n$, dan geeft `lcoeff(p,x)` de coëfficiënt van x^n.

`gcd` Het commando `gcd(p,q)` berekent de *grootste gemeenschappelijke deler* van de polynomen `p` en `q`.

11.5 Rationale functies

Een rationale functie is een quotiënt van twee polynomen. Als `R` zo'n rationale functie is, kunnen we de teller en noemer van `R` verkrijgen middels `numer(R)` respectievelijk `denom(R)`. Het commando `normal(R)` zorgt ervoor dat `R` in de vorm `p/q` komt te staan, waarbij `p` en `q` relatief priem zijn. Denk er echter aan dat dit lang niet altijd de handigste vorm is voor een rationale functie: op $(1-x^{1000})/(1-x)$ kun je `normal` beter niet toepassen. Ook combinaties (als sommen, quotiënten) van rationale functies krijgen we in de vorm p/q met `normal`.

`numer, denom`

`normal`

`factor` Met `factor(R)` worden teller en noemer afzonderlijk in factoren ontbonden.

Voorbeeldopgave

Breng de termen van

$$\frac{y}{1-x} + \frac{x}{1-y} + \frac{z}{1-xy}$$

op één noemer en bepaal de teller en noemer.

Voorbeeldsessie

```
> r := y/(1-x) + x/(1-y) + z/(1-x*y);
```

$$r := \frac{y}{1-x} + \frac{x}{1-y} + \frac{z}{1-x\,y}$$

```
> R := normal(r);
```

$$R := -\frac{y - x\,y^2 - y^2 + x\,y^3 + x - x^2\,y - x^2 + x^3\,y + z - z\,y - z\,x + z\,x\,y}{(-1+x)\,(-1+y)\,(-1+x\,y)}$$

```
> normal( r, expanded );
```

$$\frac{-y + x\,y^2 + y^2 - x\,y^3 - x + x^2\,y + x^2 - x^3\,y - z + z\,y + z\,x - z\,x\,y}{-1 + y - x\,y^2 + x - x^2\,y + x^2\,y^2}$$

```
> teller := numer(r);
```

$$teller := -y + x\,y^2 + y^2 - x\,y^3 - x + x^2\,y + x^2 - x^3\,y - z + z\,y + z\,x - z\,x\,y$$

```
> noemer := denom(r);
```

$$noemer := (-1+x)\,(-1+y)\,(-1+x\,y)$$

Toelichting

Van het commando `normal` is er ook een versie met `expanded` als tweede argument. Dit heeft hetzelfde effect als `expand(normal(r))`. Merk op dat het voor het berekenen van teller en noemer niet nodig is om eerst het commando `normal` te geven. ◇

11.6 Breuksplitsing

Ten slotte noemen we nog een opdracht die nuttig is bij het *breuksplitsen*. Stel dat `R` een rationale functie is met variabele `x`. Dan levert

`parfrac`

```
convert(R,parfrac,x)
```

de breuksplitsing van `R`.

Voorbeeldopgave

Bepaal de breuksplitsing van $\dfrac{x(x+1)(x+2)}{x^2+2}$.

Voorbeeldsessie

```
> r := x*(x+1)*(x+2)/(x^2+2);
```

$$r := \frac{x\,(x+1)\,(x+2)}{x^2+2}$$

```
> convert(r, parfrac, x);
```

$$x+3-\frac{6}{x^2+2}$$

Opgave 11.1

Ontbind $x^4 - x^2 + 1$ in reële factoren van zo laag mogelijke graad.

Opgave 11.2

Transformeer de rationale functie

$$\frac{x^4 + x^3 - x^2 - x}{x^4 - x^3 + x^2 - 5x - 2}$$

achtereenvolgens naar

$$\frac{x^4 + x^3 - x^2 - x}{(x-2)(x^3 + x^2 + 3x + 1)},$$

naar

$$\frac{x(x-1)(x+1)^2}{x^4 - x^3 + x^2 - 5x - 2},$$

en naar

$$\frac{x(x-1)(x+1)^2}{(x-2)(x^3 + x^2 + 3x + 1)}.$$

Opgave 11.3

Bepaal a, b, c zo dat

$$xa^2 + (x+1)a + xb + b + c + 1 + (c+b)x^2 = 0$$

voor alle x.